Computing Memory Terms of KdV Symbolically

L Operator

Linearity of L

```
In[1]:= L[ Plus[ x_, y__]] := Plus[L[x], L[Plus[y]]] (*Apply to each entry in a sum*)
        L[Times[x_Integer, y__]] := x L[y] (*Factor out constant integers*)
```

Unique properties of L in KdV

```
In[3]:= L[Times[I, y__]] := I L[Times[y]] (*Factor out imaginary constant*)
        L[Times[Complex[a_, b_], y__]] := (a + b I) L[Times[y]]
        (*factor out scaled imaginary constants*)
        L[Times[Power[ϵ, x_], y__]] := ϵ^x L[Times[y]] (*Factor out powers of epsilon*)
        L[Times[Power[k, x_], y__]] := k^x L[Times[y]] (*Factor out powers of k*)
```

P Operator

Linearity of P

```
In[7]:= P[ Plus[ x_, y__]] := Plus[P[x], P[Plus[y]]] (*Linearity*)
        P[Times[x_Integer, y__]] := x P[y] (*Factor out constant integers*)
```

Unique properties of P in KdV

```
In[9]:= P[Times[I, y__]] := I P[Times[y]] (*Factor out imaginary constant*)
        P[Times[Complex[a_, b_], y__]] := (a + b I) P[Times[y]]
        (*factor out scaled imaginary constants*)
        P[Times[Power[ϵ, x_], y__]] := ϵ^x P[Times[y]] (*Factor out powers of epsilon*)
        P[Times[Power[k, x_], y__]] := k^x P[Times[y]] (*Factor out powers of k*)
        P[u] := u (*Resolved modes unaffected by P*)
        P[ustar] := 0 (*Unresolved modes set to zero*)
```

Orthogonal complement of P

```
In[15]:= Q[x_] := x - P[x] (*Defined in terms of difference between identity and P*)
```

KdV Specific Terms

Convolution sums

In[116]:=
```
SetAttributes[Ck, Orderless] (*Resolved convolution sum is symmetric*)
SetAttributes[Ckstar, Orderless] (*Unresolved convolution sum is symmetric*)
Ck[Times[x_Integer, y__], z_] := x Ck[Times[y], z] (*Factor out integers*)
Ckstar[Times[x_Integer, y__], z_] := x Ckstar[Times[y], z] (*Factor out integers*)
Ck[0, x_] := 0 (*Convolution with zero is zero*)
Ckstar[0, x_] := 0 (*Convolution with zero is zero*)
```

L operator definition

Resolved modes

In[22]:=
```
L[u] := I ϵ^2 k^3 u + Ck[u, u] + Ck[u, ustar] + Ck[ustar, u] + Ck[ustar, ustar]
```

Unresolved modes

In[23]:=
```
L[ustar] := I ϵ^2 k^3 ustar + Ckstar[u, u] +
  Ckstar[u, ustar] + Ckstar[ustar, u] + Ckstar[ustar, ustar]
```

Product rule applied to convolutions

In[24]:=
```
L[Ck[x_, y_]] := Ck[L[x], y] + Ck[x, L[y]]
(*Product rule for resolved convolution*)
L[Ckstar[x_, y_]] :=
 Ckstar[L[x], y] + Ckstar[x, L[y]] (*Product rule for unresolved convolution*)
```

P operator specification and simplification

P operator applied to convolutions

In[26]:=
```
P[Ck[a_, b_]] := Ck[P[a], P[b]] (*Resolved convolution*)
P[Ckstar[a_, b_]] := Ckstar[P[a], P[b]] (*Unresolved convolution*)
```

Simplification operator (apply after expanding final result)

Interactions between resolved convolutions:

In[28]:=
```
(*Group resolved convolutions with matching arguments (both have coefficients)*)
PSimplify[Plus[Times[a_ , Ck[x_, y_]], Times[b_ , Ck[x_, z_]], w__]] :=
 PSimplify[Plus[Ck[x, a y + b z], w]]
PSimplify[Plus[Times[a_, Ck[x_, y_]], Times[b_, Ck[x_, z_]]]] :=
 Simplify[Ck[x, a y + b z]]
(*Group resolved convolutions with matching arguments (one has coefficient)*)
PSimplify[Plus[Ck[x_, y_], Times[b_ , Ck[x_, z_]], w__]] :=
 PSimplify[Plus[Ck[x, y + b z], w]]
PSimplify[Plus[Times[a_, Ck[x_, y_]], Ck[x_, z_]]] := Simplify[Ck[x, a y + z]]
(*Group resolved convolutions with
 matching arguments (neither has coefficients)*)
PSimplify[Plus[Ck[x_, y_], Ck[x_, z_], w__]] := PSimplify[Plus[Ck[x, y + z], w]]
PSimplify[Plus[Ck[x_, y_], Ck[x_, z_]]] := Simplify[Ck[x,  y + z]]
(*If terms cannot be grouped, simplify as is (both have coefficients)*)
PSimplify[Plus[Times[a_, Ck[x_, y_]], Times[b_, Ck[u_, v_]], w__]] :=
 Simplify[Plus[a Ck[x, y], b Ck[u, v], PSimplify[w]]]
PSimplify[Plus[Times[a_, Ck[x_, y_]], Times[b_, Ck[u_, v_]]]] :=
 Simplify[Plus[a Ck[x, y], b Ck[u, v]]]
(*If terms cannot be grouped, simplify as is (one has coefficient)*)
PSimplify[Plus[Times[a_, Ck[x_, y_]], Ck[u_, v_]], w__] :=
 Simplify[Plus[a Ck[x, y], Ck[u, v], PSimplify[w]]]
PSimplify[Plus[Times[a_, Ck[x_, y_]], Ck[u_, v_]]] :=
 Simplify[Plus[a Ck[x, y], Ck[u, v]]]
(*If terms cannot be grouped, simplify as is (neither has coefficients)*)
PSimplify[Plus[Ck[x_, y_], Ck[u_, v_], w__]] :=
 Simplify[Plus[Ck[x, y], Ck[u, v], PSimplify[w]]]
PSimplify[Plus[Ck[x_, y_], Ck[u_, v_]]] := Simplify[Plus[Ck[x, y], Ck[u, v]]]
(*Simplification of single terms*)
PSimplify[Times[a_, Ck[x_, y_]]] := Times[a, Ck[x, y]]
PSimplify[Ck[x_, y_]] := Ck[x, y]
```

Interactions between unresolved convolutions:

```
In[42]:= (*Group resolved convolutions with matching arguments (both have coefficients)*)
PSimplify[Plus[Times[a_ , Ckstar[x_, y_]], Times[b_ , Ckstar[x_, z_]], w__]] :=
 PSimplify[Plus[Ckstar[x, a y + b z], w]]
PSimplify[Plus[Times[a_, Ckstar[x_, y_]], Times[b_, Ckstar[x_, z_]]]] :=
 Simplify[Ckstar[x, a y + b z]]
(*Group resolved convolutions with matching arguments (one has coefficient)*)
PSimplify[Plus[Ckstar[x_, y_], Times[b_ , Ckstar[x_, z_]], w__]] :=
 PSimplify[Plus[Ckstar[x, y + b z], w]]
PSimplify[Plus[Times[a_, Ckstar[x_, y_]], Ckstar[x_, z_]]] :=
 Simplify[Ckstar[x, a y + z]]
(*Group resolved convolutions with matching
 arguments (neither has coefficients)*)
PSimplify[Plus[Ckstar[x_, y_], Ckstar[x_, z_], w__]] :=
 PSimplify[Plus[Ckstar[x, y + z], w]]
PSimplify[Plus[Ckstar[x_, y_], Ckstar[x_, z_]]] := Simplify[Ckstar[x,  y + z]]
(*If terms cannot be grouped, simplify as is (both have coefficients)*)
PSimplify[Plus[Times[a_, Ckstar[x_, y_]], Times[b_, Ckstar[u_, v_]], w__]] :=
 Simplify[Plus[a Ckstar[x, y], b Ckstar[u, v], PSimplify[w]]]
PSimplify[Plus[Times[a_, Ckstar[x_, y_]], Times[b_, Ckstar[u_, v_]]]] :=
 Simplify[Plus[a Ckstar[x, y], b Ckstar[u, v]]]
(*If terms cannot be grouped, simplify as is (one has coefficient)*)
PSimplify[Plus[Times[a_, Ckstar[x_, y_]], Ckstar[u_, v_]], w__] :=
 Simplify[Plus[a Ckstar[x, y], Ckstar[u, v], PSimplify[w]]]
PSimplify[Plus[Times[a_, Ckstar[x_, y_]], Ckstar[u_, v_]]] :=
 Simplify[Plus[a Ckstar[x, y], Ckstar[u, v]]]
(*If terms cannot be grouped, simplify as is (neither has coefficients)*)
PSimplify[Plus[Ckstar[x_, y_], Ckstar[u_, v_], w__]] :=
 Simplify[Plus[Ckstar[x, y], Ckstar[u, v], PSimplify[w]]]
PSimplify[Plus[Ckstar[x_, y_], Ckstar[u_, v_]]] :=
 Simplify[Plus[Ckstar[x, y], Ckstar[u, v]]]
(*Simplification of single terms*)
PSimplify[Times[a_, Ckstar[x_, y_]]] := Times[a, Ckstar[x, y]]
PSimplify[Ckstar[x_, y_]] := Ckstar[x, y]
```

Interactions between resolved and unresolved convolutions :

In[56]:=
```
(*Group resolved convolutions with matching arguments (both have coefficients)*)
PSimplify[Plus[Times[a_, Ck[x_, y_]], Times[b_, Ckstar[u_, v_]], w__]] :=
 Simplify[Plus[a Ck[x, y], b Ckstar[u, v], PSimplify[w]]]
PSimplify[Plus[Times[a_, Ck[x_, y_]], Times[b_, Ckstar[u_, v_]]]] :=
 Simplify[Plus[a Ck[x, y], b Ckstar[u, v]]]
(*Group resolved convolutions with matching arguments (one has coefficient)*)
PSimplify[Plus[Times[a_, Ck[x_, y_]], Ckstar[u_, v_]], w__] :=
 Simplify[Plus[a Ck[x, y], Ckstar[u, v], PSimplify[w]]]
PSimplify[Plus[Times[a_, Ck[x_, y_]], Ckstar[u_, v_]]] :=
 Simplify[Plus[a Ck[x, y], Ckstar[u, v]]]
PSimplify[Plus[Times[a_, Ckstar[x_, y_]], Ck[u_, v_]], w__] :=
 Simplify[Plus[a Ckstar[x, y], Ck[u, v], PSimplify[w]]]
PSimplify[Plus[Times[a_, Ckstar[x_, y_]], Ck[u_, v_]]] :=
 Simplify[Plus[a Ckstar[x, y], Ck[u, v]]]
(*Group resolved convolutions with matching
 arguments (neither has coefficients)*)
PSimplify[Plus[Ck[x_, y_], Ckstar[u_, v_], w__]] :=
 Simplify[Plus[Ck[x, y], Ckstar[u, v], PSimplify[w]]]
PSimplify[Plus[Ck[x_, y_], Ckstar[u_, v_]]] := Simplify[Plus[Ck[x, y], Ckstar[u, v]]]
```

Computing Terms of KdV

Full model

In[64]:= `L[u]`

Out[64]= $i\, k^3\, u\, \epsilon^2 + \mathrm{Ck}[u, u] + 2\, \mathrm{Ck}[u, \mathrm{ustar}] + \mathrm{Ck}[\mathrm{ustar}, \mathrm{ustar}]$

Markov term

In[65]:= `P[L[u]]`

Out[65]= $i\, k^3\, u\, \epsilon^2 + \mathrm{Ck}[u, u]$

t - model term

In[66]:= `PSimplify[Expand[P[L[Q[L[u]]]]]]`

Out[66]= $2\, \mathrm{Ck}[u, \mathrm{Ckstar}[u, u]]$

t^2 - model term (BCH)

In[67]:= `PSimplify[Expand[P[L[P[L[Q[L[u]]]]]]]]`

Out[67]= $2\, \left(2\, \mathrm{Ck}\left[u, \mathrm{Ckstar}\left[u, i\, k^3\, u\, \epsilon^2 + \mathrm{Ck}[u, u]\right]\right] + \mathrm{Ck}\left[i\, k^3\, u\, \epsilon^2 + \mathrm{Ck}[u, u], \mathrm{Ckstar}[u, u]\right]\right)$

t^3 – model term (BCH)

```
In[68]:= PSimplify[Expand[P[L[P[L[P[L[Q[L[u]]]]]]]]]]
```

```
Out[68]= 2 (2 Ck[u, Ckstar[u, -k^6 u ϵ^4 + ⅈ k^3 ϵ^2 Ck[u, u] + 2 Ck[u, ⅈ k^3 u ϵ^2 + Ck[u, u]]] +
            Ckstar[ⅈ k^3 u ϵ^2 + Ck[u, u], ⅈ k^3 u ϵ^2 + Ck[u, u]]] +
         4 Ck[ⅈ k^3 u ϵ^2 + Ck[u, u], Ckstar[u, ⅈ k^3 u ϵ^2 + Ck[u, u]]] +
         Ck[-k^6 u ϵ^4 + ⅈ k^3 ϵ^2 Ck[u, u] + 2 Ck[u, ⅈ k^3 u ϵ^2 + Ck[u, u]], Ckstar[u, u]])
```

t^2 – model term (Complete)

```
In[69]:= PSimplify[Expand[P[L[P[L[Q[L[u]]]]]] - P[L[Q[L[Q[L[u]]]]]]]]
```

```
Out[69]= Ck[u, -2 ⅈ k^3 ϵ^2 Ckstar[u, u] + 4 Ckstar[u, ⅈ k^3 u ϵ^2 + Ck[u, u]] -
            4 Ckstar[u, Ckstar[u, u]]] - 2 Ck[Ckstar[u, u], Ckstar[u, u]]
```

t^3 – model term (Complete)

```
In[123]:= PSimplify[Expand[P[L[P[L[P[L[Q[L[u]]]]]]]] - 2 P[L[Q[L[P[L[Q[L[u]]]]]]]] -
             2 P[L[P[L[Q[L[Q[L[u]]]]]]]] + P[L[Q[L[Q[L[Q[L[u]]]]]]]]]]
```

```
Out[123]= Ck[u, -2 k^6 ϵ^4 Ckstar[u, u] + 4 (-2 ⅈ k^3 ϵ^2 Ckstar[u, ⅈ k^3 u ϵ^2 + Ck[u, u]] +
                 3 Ckstar[u, -k^6 u ϵ^4 + ⅈ k^3 ϵ^2 Ck[u, u] + 2 Ck[u, ⅈ k^3 u ϵ^2 + Ck[u, u]]] - 2 Ckstar[u,
                   -k^6 u ϵ^4 + ⅈ k^3 ϵ^2 Ck[u, u] + 2 Ck[u, ⅈ k^3 u ϵ^2 + Ck[u, u]] + 2 Ck[u, Ckstar[u, u]]] +
                 ⅈ k^3 ϵ^2 Ckstar[u, Ckstar[u, u]] - 6 Ckstar[u, Ckstar[u, ⅈ k^3 u ϵ^2 + Ck[u, u]]] +
                 Ckstar[u, ⅈ k^3 ϵ^2 Ckstar[u, u] +
                   2 (Ckstar[u, ⅈ k^3 u ϵ^2 + Ck[u, u]] + Ckstar[u, Ckstar[u, u]])] +
                 Ckstar[ⅈ k^3 u ϵ^2 + Ck[u, u], ⅈ k^3 u ϵ^2 + Ck[u, u]] -
                 Ckstar[ⅈ k^3 u ϵ^2 + Ck[u, u], Ckstar[u, u]] + Ckstar[Ckstar[u, u], Ckstar[u, u]])] +
            6 Ck[Ckstar[u, u], ⅈ k^3 ϵ^2 Ckstar[u, u] - 2 Ckstar[u, ⅈ k^3 u ϵ^2 + Ck[u, u]] +
               2 Ckstar[u, Ckstar[u, u]]]
```